ArduinoをPS4 DualShockコントローラで動かす

# 目的
ArduinoをPS4 DualShockコントローラで動かした際の備忘録です。

# 準備

Arduino Uno
[PS4 DualShockコントローラ](https://www.jp.playstation.com/ps4/peripheral/cuhzct2j13.html)
[USBホストシールド 2.0 for Arduino ](https://www.amazon.co.jp/dp/B007YNG6RW/ref=cm_sw_r_tw_dp_U_x_vyelDbWDEH3EP)
[エレコム Bluetooth USBアダプタ](https://www.amazon.co.jp/dp/B00J84V7OM/ref=cm_sw_r_tw_dp_U_x_PzelDb26TK80R)

USBホストシールドは入手後にはんだブリッジを行います。

[USB ホストシールド](https://ht-deko.com/arduino/shield_usbhost.html)
> そしてこの製品にはハンダブリッジしないと動かないというとんでもない罠があるのです。
> 5V で使う場合

[USBホストシールド](http://leavebehind.iobb.net/wordpress/2018/01/06/usbhostshield/)
> USBホストシールド 2.0 for Arduino (compatible with Google Android ADK)[amazon.co.jp]を購入して半田付けせんと動かんかったちゅー話。

[USB_Host_Shield_2.0](https://github.com/felis/USB_Host_Shield_2.0)ライブラリ入手してArduinoIDEへインストールします。

# コード

USB_Host_Shield_2.0ライブラリのPS4BTのサンプルコードを使用します。

```rb:PS4BT
#include <PS4BT.h>
#include <usbhub.h>

// Satisfy the IDE, which needs to see the include statment in the ino too.
#ifdef dobogusinclude
#include <spi4teensy3.h>
#endif
#include <SPI.h>

USB Usb;
//USBHub Hub1(&Usb); // Some dongles have a hub inside
BTD Btd(&Usb); // You have to create the Bluetooth Dongle instance like so

/* You can create the instance of the PS4BT class in two ways */
// This will start an inquiry and then pair with the PS4 controller - you only have to do this once
// You will need to hold down the PS and Share button at the same time, the PS4 controller will then start to blink rapidly indicating that it is in pairing mode
PS4BT PS4(&Btd, PAIR);

// After that you can simply create the instance like so and then press the PS button on the device
//PS4BT PS4(&Btd);

bool printAngle, printTouch;
uint8_t oldL2Value, oldR2Value;

void setup() {
  Serial.begin(115200);
#if !defined(__MIPSEL__)
  while (!Serial); // Wait for serial port to connect - used on Leonardo, Teensy and other boards with built-in USB CDC serial connection
#endif
  if (Usb.Init() == -1) {
    Serial.print(F("\r\nOSC did not start"));
    while (1); // Halt
  }
  Serial.print(F("\r\nPS4 Bluetooth Library Started"));
}
void loop() {
  Usb.Task();
//delay(1);
//Serial.println(PS4.connected());

  if (PS4.connected()) {
    if (PS4.getAnalogHat(LeftHatX) > 137 || PS4.getAnalogHat(LeftHatX) < 117 || PS4.getAnalogHat(LeftHatY) > 137 || PS4.getAnalogHat(LeftHatY) < 117 || PS4.getAnalogHat(RightHatX) > 137 || PS4.getAnalogHat(RightHatX) < 117 || PS4.getAnalogHat(RightHatY) > 137 || PS4.getAnalogHat(RightHatY) < 117) {
      Serial.print(F("\r\nLeftHatX: "));
      Serial.print(PS4.getAnalogHat(LeftHatX));
      Serial.print(F("\tLeftHatY: "));
      Serial.print(PS4.getAnalogHat(LeftHatY));
      Serial.print(F("\tRightHatX: "));
      Serial.print(PS4.getAnalogHat(RightHatX));
      Serial.print(F("\tRightHatY: "));
      Serial.print(PS4.getAnalogHat(RightHatY));
    }

    if (PS4.getAnalogButton(L2) || PS4.getAnalogButton(R2)) { // These are the only analog buttons on the PS4 controller
      Serial.print(F("\r\nL2: "));
      Serial.print(PS4.getAnalogButton(L2));
      Serial.print(F("\tR2: "));
      Serial.print(PS4.getAnalogButton(R2));
    }
    if (PS4.getAnalogButton(L2) != oldL2Value || PS4.getAnalogButton(R2) != oldR2Value) // Only write value if it's different
      PS4.setRumbleOn(PS4.getAnalogButton(L2), PS4.getAnalogButton(R2));
    oldL2Value = PS4.getAnalogButton(L2);
    oldR2Value = PS4.getAnalogButton(R2);

    if (PS4.getButtonClick(PS)) {
      Serial.print(F("\r\nPS"));
      PS4.disconnect();
    }
    else {
      if (PS4.getButtonClick(TRIANGLE)) {
        Serial.print(F("\r\nTriangle"));
        PS4.setRumbleOn(RumbleLow);
      }
      if (PS4.getButtonClick(CIRCLE)) {
        Serial.print(F("\r\nCircle"));
        PS4.setRumbleOn(RumbleHigh);
      }
      if (PS4.getButtonClick(CROSS)) {
        Serial.print(F("\r\nCross"));
        PS4.setLedFlash(10, 10); // Set it to blink rapidly
      }
      if (PS4.getButtonClick(SQUARE)) {
        Serial.print(F("\r\nSquare"));
        PS4.setLedFlash(0, 0); // Turn off blinking
      }

      if (PS4.getButtonClick(UP)) {
        Serial.print(F("\r\nUp"));
        PS4.setLed(Red);
      } if (PS4.getButtonClick(RIGHT)) {
        Serial.print(F("\r\nRight"));
        PS4.setLed(Blue);
      } if (PS4.getButtonClick(DOWN)) {
        Serial.print(F("\r\nDown"));
        PS4.setLed(Yellow);
      } if (PS4.getButtonClick(LEFT)) {
        Serial.print(F("\r\nLeft"));
        PS4.setLed(Green);
      }

      if (PS4.getButtonClick(L1))
        Serial.print(F("\r\nL1"));
      if (PS4.getButtonClick(L3))
        Serial.print(F("\r\nL3"));
      if (PS4.getButtonClick(R1))
        Serial.print(F("\r\nR1"));
      if (PS4.getButtonClick(R3))
        Serial.print(F("\r\nR3"));

      if (PS4.getButtonClick(SHARE))
        Serial.print(F("\r\nShare"));
      if (PS4.getButtonClick(OPTIONS)) {
        Serial.print(F("\r\nOptions"));
        printAngle = !printAngle;
      }
      if (PS4.getButtonClick(TOUCHPAD)) {
        Serial.print(F("\r\nTouchpad"));
        printTouch = !printTouch;
      }

      if (printAngle) { // Print angle calculated using the accelerometer only
        Serial.print(F("\r\nPitch: "));
        Serial.print(PS4.getAngle(Pitch));
        Serial.print(F("\tRoll: "));
        Serial.print(PS4.getAngle(Roll));
      }

      if (printTouch) { // Print the x, y coordinates of the touchpad
        if (PS4.isTouching(0) || PS4.isTouching(1)) // Print newline and carriage return if any of the fingers are touching the touchpad
          Serial.print(F("\r\n"));
        for (uint8_t i = 0; i < 2; i++) { // The touchpad track two fingers
          if (PS4.isTouching(i)) { // Print the position of the finger if it is touching the touchpad
            Serial.print(F("X")); Serial.print(i + 1); Serial.print(F(": "));
            Serial.print(PS4.getX(i));
            Serial.print(F("\tY")); Serial.print(i + 1); Serial.print(F(": "));
            Serial.print(PS4.getY(i));
            Serial.print(F("\t"));
          }
        }
      }
    }
  }
}
```

# テスト

PS4コントローラの「PSボタン」「SHAREボタン」を同時押してコントローラとArduinoに繋いたBTドングルとベアリングする。
コントローラの入力が認識されればOK。

<blockquote class="twitter-tweet" data-lang="en"><p lang="ja" dir="ltr">arduino usb shield + PS4コントローラ <a href="https://t.co/pvz0qmWHwX">pic.twitter.com/pvz0qmWHwX</a></p>&mdash; st (@st17890027) <a href="https://twitter.com/st17890027/status/1150220976035991554?ref_src=twsrc%5Etfw">July 14, 2019</a></blockquote> <script async src="https://platform.twitter.com/widgets.js" charset="utf-8"></script> 



# Error対策

```
$ OSC did not start
```
--> ArduinoとUSB Host Shieldの接続を見直す必要あり

--

互換品のDualShock 4 コントローラで上記の通りペアリングを行った記事が見付かりませんでした。
電子工作用途であっても正規品DualShock 4を利用した方が良いかもしれません。（BT通信フォーマットの差分が影響している？）

# 参考
[PS4コントローラのスティックでarduinoを制御する](https://qiita.com/igapon1/items/e28ea499d85fa658c919)
[Arduino Uno R3にPS4コントローラーをBluetoothで繋いでみた](http://high-sider.hateblo.jp/entry/2015/02/14/233211)
[robo8080のブログ Arduino + USBホストシールドの実験](https://blog.goo.ne.jp/roboz80/e/9d2fc6e1266932f7ace682b43393217b)
[Revision 2.0 of USB Host Library for Arduino.](https://github.com/felis/USB_Host_Shield_2.0)
[USB_Host_Shield_2.0](https://github.com/felis/USB_Host_Shield_2.0)
[USB ホストシールド ht-deko.com](https://ht-deko.com/arduino/shield_usbhost.html)
[USB ホストシールド leavebehind.iobb.net](http://leavebehind.iobb.net/wordpress/2018/01/06/usbhostshield/)